
Metaattribute definieren in einem INTERLIS Modell 
 
Dieses Dokument ist ein Entwurf. 
 
Dieses Dokument beschreibt die Definition von Metaattributen als Teil eines 
INTERLIS Modells (ili-Datei). 

Einleitung 
Metaattribute erlauben eine Ergänzung der INTERLIS Modelle mit zusätzlichen 
Angaben, die in der aktuellen Spezifikation von INTERLIS nicht vorgesehen sind. 
Im Dokument zum INTERLIS 2-Metamodell steht dazu: 

Einem Metaelement können kompositorisch Metaattribute zugeordnet sein (Klasse 
"MetaAttribut"). Metaattribute sind weder durch die Sprache INTERLIS noch durch das 
Metamodell genauer definiert. Sie dienen insbesondere dazu, dass Informationen, die über 
INTERLIS hinausgehen, als Bestandteil der Modelldaten beigefügt werden können. Metaattribute 
haben einen Namen (Attribut "Name"), der innerhalb der Metaattribute desselben Metaelementes 
eindeutig sein muss, und den beschreibenden Wert (Attribut "Value"). 

Im vorliegenden Dokument wird eine Syntax beschrieben, so dass solche 
Metaattribute direkt als Teil des Modells formuliert werden können (im Gegensatz 
zum Metamodell, wo definiert wird, wie Metaattribute als Daten definiert bzw. in XML 
transferiert werden). 

Konzept 
Um die bestehende Sprachdefinition nicht ändern zu müssen, werden die 
Metaattribute innerhalb von Zeilenkommentar (Text auf einer Zeile ab '!!') formuliert. 
Um Metaattribute von echtem Kommentar unterscheiden zu können, ist als drittes 
Zeichen zwingend das Zeichen '@', d.h. ein Zeilenkommentar der Metaattribute 
enthält, beginnt mit '!!@'. 
Zu welchen Metaelement ein solches Metaattribut gehört, ergibt sich aus der Position 
des Zeilenkommentars. Das Metaattribut bezieht sich auf das unmittelbar folgende 
Sprachkonstrukt (z.B. ModelDef oder ClassDef). 

Syntax 
MetaAttribute = '!!@' Name '=' Value { ';' [ Name '=' Value ] }. 
Value = Name | String.  
Name = (*  ~('\t' | '\f' | '\r' | '\n' | ' ' | '=' | ';' | ',' | '"') *). 
String = '"' { Escape | ~( '"' | '\\' ) } '"'. 
Escape = '\\' ( '"' | '\\' | 'u' HexDigit HexDigit HexDigit HexDigit ). 
HexDigit = '0' .. '9' | 'a' .. 'f' | 'A' .. 'F'. 
 
Als Zusatz zur EBNF Definition im INTERLIS 2-Referenzhandbuch (Kap. 2.1) gilt: 
~A ist die Negation 
A .. B bezeichnet einen Bereich von Zeichen 
'\t' bezeichnet das Tabulatorzeichen (Unicode Name: <CHARACTER 

TABULATION>) 
'\f' bezeichnet das Seitenvorschubzeichen (<FORM FEED (FF)>) 
'\r' bezeichnet das Wagenrücklaufzeichen (<CARRIAGE RETURN (CR)>) 
'\n' bezeichnet das Zeilenvorschubzeichen (<LINE FEED (LF)>) 

Positionierung innerhalb der INTERLIS Sprache 
In den folgenden Ausschnitten aus Syntaxregeln aus dem INTERLIS 
Referenzhandbuch wird definiert, an welcher Stelle Zeilenkommentare mit 
Metaattributen vorkommen dürfen. Zeilenkommentare mit Metaattributen an anderen 
Stellen werden ignoriert. 
ModelDef = {MetaAttribute} [ 'CONTRACTED' ]  
           [ 'TYPE' | 'REFSYSTEM' | 'SYMBOLOGY' ] 
           'MODEL' Model-Name ... 
TopicDef = {MetaAttribute} [ 'VIEW' ] 'TOPIC' Topic-Name ... 
ClassDef =  {MetaAttribute} 'CLASS' Class-Name ... 
StructureDef =  {MetaAttribute} 'STRUCTURE' Structure-Name ... 
AttributeDef = {MetaAttribute} [ [ 'CONTINUOUS' ] 'SUBDIVISION' ]  
                Attribute-Name ... 
AssociationDef =  {MetaAttribute} 'ASSOCIATION' [ Association-Name ] ... 
RoleDef = {MetaAttribute} Role-Name ... 
DomainDef = 'DOMAIN' { {MetaAttribute} Domain-Name ... 
EnumElement = {MetaAttribute} EnumElement-Name ... 
LineFormTypeDef = 'LINE' 'FORM' { {MetaAttribute} LineFormType-Name  ... 
UnitDef = 'UNIT' { {MetaAttribute} Unit-Name ... 
MetaDataBasketDef = {MetaAttribute} ( 'SIGN' | 'REFSYSTEM' )  
              'BASKET' Basket-Name ... 
ParameterDef = {MetaAttribute} Parameter-Name ... 
RunTimeParameterDef = 'PARAMETER' { {MetaAttribute}  
               RunTimeParameter-Name ... 
ConstraintDef = {MetaAttribute}  ( MandatoryConstraint | ... 
FunctionDef = {MetaAttribute} 'FUNCTION' Function-Name ... 
ViewDef = {MetaAttribute} 'VIEW' View-Name ... 
GraphicDef = {MetaAttribute} 'GRAPHIC' Graphic-Name ... 
Metaattribute zu einem TopicDef werden in den Schemadaten (Daten/Objekte 
gemäss Metamodell) dem SubModel zugeordnet (und nicht der DataUnit). 
Metaattribute zu einem ConstraintDef sind als Schemadaten nicht möglich (weil 
Constraints gemäss Metamodell keine OID haben bzw. kein MetaElement sind). 

Beispiel 
Im folgenden Beispiel wäre "ili.charset" der Metaattribut Name und "ISO-8859-1" der 
Metaattribut Wert.  
INTERLIS 2.3; 
!!@ ili.charset=ISO-8859-1 
MODEL Example 
  AT "mailto:noreply@localhost" 
  VERSION "2009-01-09"  = 
END Example. 
 
Das Metaattribut würde einem Objekt der Klasse "Model" aus dem Metamodell mit 
dem Namen "Example" zugeordnet. 
 
<IlisMeta07.ModelData BID="MODEL.Example"> 
 
  <IlisMeta07.ModelData.Model TID="Example"> 
    <Name>Example</Name> 
    <iliVersion>2.3</iliVersion> 
    <Contracted>false</Contracted> 
    <Kind>NormalM</Kind> 
    <At>mailto:noreply@localhost</At> 
    <Version>2009-01-09</Version> 
  </IlisMeta07.ModelData.Model> 
 
  <IlisMeta07.ModelData.MetaAttribute TID="Example.METAOBJECT.ili.charset"> 

    <Name>ili.charset</Name> 
    <Value>ISO-8859-1</Value> 
    <MetaElement REF="Example" /> 
  </IlisMeta07.ModelData.MetaAttribute> 
 
</IlisMeta07.ModelData> 

